
Chapter 8 Further Applications of Integration

■ 8.1 Arc Length and Surface Area

Students should read Section 8.1 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

■ 8.1.1 Arc Length

The integrals for calculating arc length and surface area are generally difficult to do by hand. Thus, *Mathematica* is the appropriate tool for evaluating these integrals.

If y is a function of x , that is, $y = f(x)$, and $f'(x)$ exists and is continuous on $[a, b]$, then the arc length of the graph of $f(x)$ over the interval $[a, b]$ is

$$L = \int_a^b \sqrt{1 + f'(x)^2} \, dx$$

If x is a function of y , that is, $x = g(y)$, and $g'(y)$ exists and is continuous on $[c, d]$, then the arc length of the graph of $g(y)$ over the interval $[c, d]$ is

$$L = \int_c^d \sqrt{f'(y)^2 + 1} \, dy$$

Example 8.1. Estimate the arc length of $y = \frac{1}{x}$ over the interval $[1, 2]$.

Solution: Finding the arc length of this simple rational function by hand is virtually impossible. This is because $f'(x) = -\frac{1}{x^2}$ and

thus the arc length integral is $L = \int_1^2 \sqrt{1 + \frac{1}{x^4}} \, dx$, which cannot be evaluated in terms of elementary functions, as the following answer illustrates.

$$\text{In[68]:= } \int_1^2 \sqrt{1 + \frac{1}{x^4}} \, dx$$

$$\text{Out[68]:= } \frac{\sqrt{2} \pi \text{Gamma}\left[\frac{7}{4}\right]}{3 \text{Gamma}\left[\frac{5}{4}\right]} - \frac{1}{2} \text{Hypergeometric2F1}\left[-\frac{1}{2}, -\frac{1}{4}, \frac{3}{4}, -16\right]$$

However, there are numerical techniques that we can use. For example, the *Mathematica* command **NIntegrate** uses sophisticated algorithms to give us a good estimate for this definite integral:

$$\text{In[69]:= } \text{NIntegrate}\left[\sqrt{1 + \frac{1}{x^4}}, \{x, 1, 2\}\right]$$

$$\text{Out[69]= } 1.13209$$

A more elementary method of estimating this arc length is Simpson's Rule as shown in Section 7.1 of this text.

```
In[70]:= Clear[f, a, b, n]
SIMP[a_, b_, n_, f_] :=
(1/3) Sum[f[a + (2 i - 2) (b - a) / n] + 4 f[a + (2 i - 1) (b - a) / n] +
f[a + 2 i (b - a) / n], {i, 1, n/2}] (b - a) / n
```

```
In[72]:= f[x_] :=  $\sqrt{1 + \frac{1}{x^4}}$ 
```

```
TableForm[Table[{n, N[SIMP[1, 2, n, f]]}, {n, 10, 100, 10}],
TableHeadings -> {{}, {"n", "Sn"}]
```

Out[73]/TableForm=

n	S _n
10	1.1321
20	1.13209
30	1.13209
40	1.13209
50	1.13209
60	1.13209
70	1.13209
80	1.13209
90	1.13209
100	1.13209

Thus, we see that Simpson's Rule gives us as accurate an estimate of the arc length, as does the **NIntegrate** command for n as small as 20.

Example 8.2. Consider the ellipse whose equation is given by

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$


Assume that $a > b$. Find the arc length of the upper half of the ellipse.

Solution: To plot the ellipse for various values of a and b , we define a plotting command **plot[a,b]** as follows.

```
In[74]:= Clear[a, b, x, y, eq, plot]
eq[x_, y_, a_, b_] :=  $\frac{x^2}{a^2} + \frac{y^2}{b^2} - 1$ 
plot[a_, b_] := ContourPlot[eq[x, y, a, b] == 0, {x, -a, a}, {y, -b, b},
AspectRatio -> Automatic, Axes -> True, Frame -> False]
```

Here is a plot of the ellipse for $a = 2$ and $b = 3$.

In[77]:= `plot[2, 3]`



On the upper half of the ellipse, we have $y \geq 0$. Thus, we can solve for y and take the positive solution. We will denote this positive solution as a function of x , a , and b .

In[78]:= `sol = Solve[$\frac{x^2}{a^2} + \frac{y^2}{b^2} == 1, y]$;`

`f[x_, a_, b_] = sol[[2, 1, 2]]`

Out[79]=
$$\frac{b \sqrt{a^2 - x^2}}{a}$$

Clearly, the domain of f is $[-a, a]$. The natural thing to do would be to evaluate the integral $\int_{-a}^a \sqrt{1 + (f'(x))^2} dx$. Try this yourself, but be prepared to wait awhile. Moreover, *Mathematica* will give the following output:

If $\left[\text{Im}[a] == 0 \ \&\& \ \left(a \ \text{Im}\left[\frac{1}{\sqrt{-a^2 + b^2}} \right] \geq 1 \ \|\ \right. \right.$

$\left. \left. 1 + a \ \text{Im}\left[\frac{1}{\sqrt{-a^2 + b^2}} \right] \leq 0 \ \|\ a \ \text{Im}\left[\frac{1}{\sqrt{-a^2 + b^2}} \right] == 0 \ \|\ a \ \text{Re}\left[\frac{1}{\sqrt{-a^2 + b^2}} \right] \neq 0 \right) \right],$

$\frac{2 (-a)^{3/2} \sqrt{b^2} \text{EllipticE}\left[1 - \frac{b^2}{a^2}\right] \text{Sign}[a]}{\sqrt{-a b^2}}, \text{Integrate}\left[\sqrt{1 + \frac{b^2 x^2}{a^4 - a^2 x^2}}, \{x, -a, a\}, \right.$

$\left. \left. \text{Assumptions} \rightarrow (\text{Re}[b] == 0 \ \&\& \ \text{Re}[a] \neq 0 \ \&\& \ \text{Im}[a] == 0 \ \&\& \ \text{Im}[b] \neq 0) \ \|\ \text{Im}[a] < 0 \ \|\ \text{Im}[a] > 0 \right] \right]$

To understand this output, let us make a change of variable $x = a \sin t$. Then the integral becomes (verify this)

$$\int_{-a}^a \sqrt{1 + (f'(x))^2} dx = a \int_{-\pi/2}^{\pi/2} \sqrt{1 + \frac{b^2 \sin^2 t}{a^2 \cos^2 t}} \cos t dt$$

The latter integral can be expressed as

$$2a \int_0^{\pi/2} \sqrt{1 + \frac{b^2 \sin^2 t}{a^2 \cos^2 t}} \cos t \, dt = 2a \int_0^{\pi/2} \sqrt{\cos^2 t + (b^2/a^2) \sin^2 t} \, dt = 2a \int_0^{\pi/2} \sqrt{1 - c^2 \sin^2 t} \, dt,$$

where $c = \sqrt{1 - (b/a)^2}$ and we have used the identity $\cos^2 t = 1 - \sin^2 t$.

To simplify our notation, let us define the integrand in the preceding far left integral as

$$\text{In[80]:= } \mathbf{g[t_, a_, b_]} = \sqrt{1 - (1 - (b/a)^2) (\text{Sin}[t])^2}$$

$$\text{Out[80]= } \sqrt{1 - \left(1 - \frac{b^2}{a^2}\right) \text{Sin}[t]^2}$$

Here are some values of the arc length of the upper half of the ellipse.

$$\text{In[81]:= } \mathbf{TableForm}\left[\mathbf{Table}\left[2a \int_0^{\pi/2} \mathbf{g}[t, a, b] \, dt, \{a, 1, 3\}, \{b, 1, 3\}\right], \mathbf{TableHeadings} \rightarrow \{\{\mathbf{"a=1"}, \mathbf{"a=2"}, \mathbf{"a=3"}\}, \{\mathbf{"b=1"}, \mathbf{"b=2"}, \mathbf{"b=3"}\}\right]$$

Out[81]/TableForm=

	b=1	b=2	b=3
a=1	π	$2 \text{EllipticE}[-3]$	$2 \text{EllipticE}[-8]$
a=2	$4 \text{EllipticE}\left[\frac{3}{4}\right]$	2π	$4 \text{EllipticE}\left[-\frac{5}{4}\right]$
a=3	$6 \text{EllipticE}\left[\frac{8}{9}\right]$	$6 \text{EllipticE}\left[\frac{5}{9}\right]$	3π

Observe that we obtain exact values for the arc length when $a = b$. Can you explain why?

The approximate values of the numbers appearing in the preceding table are as follows:

$$\text{In[82]:= } \mathbf{TableForm}\left[\mathbf{N}\left[\mathbf{Table}\left[2a \int_0^{\pi/2} \mathbf{g}[t, a, b] \, dt, \{a, 1, 3\}, \{b, 1, 3\}\right], 10\right], \mathbf{TableHeadings} \rightarrow \{\{\mathbf{"a=1"}, \mathbf{"a=2"}, \mathbf{"a=3"}\}, \{\mathbf{"b=1"}, \mathbf{"b=2"}, \mathbf{"b=3"}\}\right]$$

Out[82]/TableForm=

	b=1	b=2	b=3
a=1	3.141592654	4.844224110	6.682446610
a=2	4.844224110	6.283185307	7.932719795
a=3	6.682446610	7.932719795	9.424777961

NOTE: The integral $\int \sqrt{1 - c^2 \sin^2 t} \, dt$ is known as an *elliptic integral*. It is very useful in mathematics and has many applications. In *Mathematica*, it is denoted by **Elliptic[t,c^2]**. The command **Elliptic[x,m]** gives $\int_0^x \sqrt{1 - m \sin^2 t} \, dt$, while **Elliptic[m]** gives $\int_0^{\pi/2} \sqrt{1 - m \sin^2 t} \, dt$.

■ 8.1.2 Surface Area

If $f'(x)$ exists and is continuous on $[a, b]$, then the surface area of revolution obtained by rotating the graph of $f(x)$ about the x -axis for $a \leq x \leq b$ is

$$S = 2\pi \int_a^b f(x) \sqrt{1 + [f'(x)]^2} dx$$

Similarly, if $x = g(y)$ and $g'(y)$ exists and is continuous on $[c, d]$, then the surface area of revolution obtained by rotating $g(y)$ about the y -axis for $c \leq y \leq d$ is

$$S = 2\pi \int_c^d g(y) \sqrt{[g'(y)]^2 + 1} dy$$

Again, evaluating these complicated integrals is what *Mathematica* does best, as the following examples illustrate.

Example 8.3. Determine the surface area of revolution obtained by rotating the region under $y = e^{-x}$ along the interval $[0, 2]$ about the x -axis.

Solution: We calculate

```
In[83]:= Clear[f, x]
```

```
f[x_] := e-x
```

$$S = 2\pi \int_0^2 f[x] \sqrt{1 + f'[x]^2} dx$$


$$\text{Out[85]} = \frac{1}{2} \pi \left(4 + 2\sqrt{2} - \frac{2}{\sqrt{1+e^4}} - \frac{2}{e^4 \sqrt{1+e^4}} + \text{Log}[3 + 2\sqrt{2}] - \text{Log}[2 + e^4 + 2\sqrt{1+e^4}] \right)$$

```
In[86]:= N[%]
```

```
Out[86]= 6.35887
```

Here is the corresponding surface of revolution (rotated 90° about the y -axis):

```
In[87]:= RevolutionPlot3D[E-x, {x, 0, 2}]
```



NOTE: Observe that in this case *Mathematica* was able to find an anti-derivative of the integrand. However, not all integrals of this form can be evaluated analytically as the next example illustrates.

Example 8.4. Determine the surface area of revolution obtained by rotating the region under $y = \tan x$ along the interval $[0, \frac{\pi}{4}]$ about the x -axis.

Solution: As in the previous example, we evaluate

```
In[88]:= Clear[f, x]
         f[x_] := Tan[x]
         NIntegrate[2 π f[x] √(1 + f'[x]^2), {x, 0, Pi/4}]
```


Out[90]= 3.83908

To appreciate the complexity of the integral and understand why we used the command **NIntegrate**, we advise the reader to define the anti-derivative **F[t]** below and evaluate **F[π/4]** (be prepared to wait awhile).

```
In[91]:= F[t_] := Integrate[f[x] √(1 + f'[x]^2), {x, 0, t}]
```

Here is the corresponding surface of revolution:

```
In[92]:= RevolutionPlot3D[Tan[x], {x, 0, Pi/4}]
```




■ Exercises

In Exercises 1 and 2, calculate the arc length of the given function over the given interval:

1. $y = x^4$, over $[1, 2]$ 2. $y = \sin x$, over $\left[0, \frac{\pi}{2}\right]$

3. Calculate the arc length of the astroid $x^{2/3} + y^{2/3} = 1$. Below is a plot of its graph. Hint: By symmetry it suffices to calculate only the portion in the first quadrant.

In[93]:= ContourPlot[(x^2)^(1/3) + (y^2)^(1/3) == 1, {x, -1, 1}, {y, -1, 1}]



4. Show that the circumference of the unit circle is 2π by calculating its arc length. Use the fact that the equation of the unit circle is given by $x^2 + y^2 = 1$.

In Exercises 5 through 7, compute the surface area of the given functions rotated about the x -axis over the given intervals:

5. $y = x^3 + \frac{1}{x}$, over $[1, 4]$ 6. $y = (4 - x^{2/3})^{2/3}$ over $[0, 8]$ 7. $y = \cos x$, over $[0, \pi]$

8. Show that the surface area of the unit sphere is 4π by rotating the top half of the unit circle $x^2 + y^2 = 1$ about the x -axis.

■ 8.2 Center of Mass

Students should read Section 8.3 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

A *lamina* is a thin plate whose mass is distributed throughout a region in the plane. Suppose a lamina has a constant density ρ and that the lamina occupies a region in the plane under the graph of a continuous function f over the interval $[a, b]$, where $f(x) \geq 0$ for all x .

The *mass* of the lamina is given by

$$M = \rho \int_a^b f(x) \, dx$$

Then the *moments* of the lamina with respect to x -axis and y -axis are denoted by M_x and M_y and are defined by

$$M_x = \frac{1}{2} \rho \int_a^b [f(x)]^2 \, dx$$

$$M_y = \rho \int_a^b x f(x) \, dx$$

The *center of mass* (also called the *centroid*) of the lamina is defined to be (\bar{x}, \bar{y}) , where

$$\bar{x} = \frac{M_y}{M} \quad \text{and} \quad \bar{y} = \frac{M_x}{M}$$

NOTE: If the lamina described above as a density ρ that continuously depends on x , that is, if $\rho = \rho(x)$ for x in the interval $[a, b]$, then the moments, the total mass, and the center of mass are given by

$$M = \int_a^b \rho(x) f(x) \, dx$$

$$M_x = \frac{1}{2} \int_a^b \rho(x) [f(x)]^2 dx$$

$$M_y = \int_a^b x \rho(x) f(x) dx$$

$$\bar{x} = \frac{M_y}{M} \text{ and } \bar{y} = \frac{M_x}{M}$$

Example 8.5. Suppose a lamina lies underneath the graph of $y = 16 - x^2$ and over the interval $[-4, 4]$.

- a) Assume the density of the lamina is $\rho = 3$. Find the mass, moments, and the center of mass of the lamina.
 b) Assume the density of the lamina is $\rho = \frac{x}{2} + 2$. Find the mass, moments, and the center of mass of the lamina.

Solution:

a) We use the above formulas with $\rho = 3$:

$$\text{In[94]:= } f[x_] = 16 - x^2$$

$$\text{Out[94]= } 16 - x^2$$

The mass is given by

$$\text{In[95]:= } M = 3 \int_{-4}^4 f[x] dx$$

$$\text{Out[95]= } 256$$

The moment with respect to the x -axis is

$$\text{In[96]:= } M_x = (3 / 2) \int_{-4}^4 (f[x])^2 dx$$

$$\text{Out[96]= } \frac{8192}{5}$$

The moment with respect to the y -axis is

$$\text{In[97]:= } M_y = 3 \int_{-4}^4 x f[x] dx$$

$$\text{Out[97]= } 0$$

The coordinates for the center of mass are

$$\text{In[98]:= } \mathbf{xbar} = M_y / M$$

$$\mathbf{ybar} = M_x / M$$

$$\text{Out[98]= } 0$$


$$\text{Out[99]= } \frac{32}{5}$$

Observe that the region of the lamina is symmetric with respect to the y -axis. Hence, the fact that $\bar{x} = 0$ is also clear from the fact that the density is a constant.

Below is the plot of the lamina and its center of mass:


```
In[100]:= plot1 = Plot[f[x], {x, -4, 4}, Filling -> Axis];
plot2 = ListPlot[{{xbar, ybar}}, PlotStyle -> {PointSize[0.02], Red}];
Show[plot1, plot2]
```

Out[102]=



b) Here, $\rho = x + 4$. With the above notation we have

```
In[103]:= f[x_] = 16 - x^2
rho[x_] = x/2 + 2
```

Out[103]= $16 - x^2$

Out[104]= $2 + \frac{x}{2}$

The mass is

```
In[105]:= Mv = Integrate[rho[x] f[x], {x, -4, 4}]
```

Out[105]= $\frac{512}{3}$

The moment with respect to the x -axis is

```
In[106]:= Mxv = (1/2) Integrate[rho[x] (f[x])^2, {x, -4, 4}]
```

Out[106]= $\frac{16384}{15}$

The moment with respect to the y -axis is

```
In[107]:= Myv = Integrate[rho[x] x f[x], {x, -4, 4}]
```

Out[107]= $\frac{2048}{15}$

The coordinates for the center of mass are


```
In[108]:= xbarv = Myv / M
          ybarv = Mxv / M
```

```
Out[108]= 8
           15
```

```
Out[109]= 64
           15
```

Here is a plot of the lamina showing the center of masses with the uniform density of $\rho = 3$ and variable density of $\rho = \frac{x}{2} + 2$ represented by the red and green dots, respectively.

```
In[110]:= plot3 = ListPlot[{{ xbarv, ybarv }}, PlotStyle -> {Green, PointSize[.02]}];
          Show[plot1, plot2, plot3]
```



NOTE: Observe that the center of mass with variable density (green dot) is shifted to the right, as expected, since the density is more weighted to the right.

Example 8.6. Suppose a lamina covers the top half of the ellipse

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

- Assume the density of the lamina is $\rho = 1$. Find the mass, moments and the center of mass of the lamina.
- Assume the density of the lamina is $\rho = e^{-x}$. Find the mass, moments and the center of mass of the lamina.

Solution: To distinguish between the uniform and variable density cases in parts a) and b), respectively, we attach the letter **u** and **v** to the notation in this solution. Thus, **Mu** will be the mass corresponding to the uniform density while **Mv** is the mass corresponding the variable density.

a) We solve the equation of the ellipse for y:

```
In[112]:= Clear[a, b, x, y]
```

```
sol = Solve[ $\frac{x^2}{a^2} + \frac{y^2}{b^2} == 1, y]$ 
```

```
Out[113]= {{y -> - $\frac{b \sqrt{a^2 - x^2}}{a}$ }, {y ->  $\frac{b \sqrt{a^2 - x^2}}{a}$ }}
```

In the top half of the ellipse, we have $y \geq 0$. Thus, we take the second solution, simplify, and define it as a function of x , a , and b

$$\text{In[114]:= } \mathbf{fa[x_, a_, b_]} := \mathbf{b} \sqrt{1 - \frac{\mathbf{x}^2}{\mathbf{a}^2}}$$

Let the mass, the moment with respect to the x -axis, the moment with respect to the y -axis, and the center of mass be denoted by $M(a, b)$, $M_x(a, b)$, $M_y(a, b)$, and $(\bar{x}(a, b), \bar{y}(a, b))$, respectively. We now compute these quantities assuming $\rho = 1$.

In[115]:= Clear[a, b, Mua, Mxua, Myua, xbaru, ybaru]

$$\mathbf{Mua[a_, b_]} = \int_{-a}^a \mathbf{fa[x, a, b]} \, dx$$

$$\mathbf{Mxua[a_, b_]} = (1/2) \int_{-a}^a (\mathbf{fa[x, a, b]})^2 \, dx$$

$$\mathbf{Myua[a_, b_]} = \int_{-a}^a \mathbf{x fa[x, a, b]} \, dx$$

$$\text{Out[116]= } \frac{a b \pi}{2}$$

$$\text{Out[117]= } \frac{2 a b^2}{3}$$

$$\text{Out[118]= } 0$$

$$\text{In[119]:= } \mathbf{xbaru[a_, b_]} = \frac{\mathbf{Myua[a, b]}}{\mathbf{Mua[a, b]}}$$

$$\mathbf{ybaru[a_, b_]} = \frac{\mathbf{Mxua[a, b]}}{\mathbf{Mua[a, b]}}$$

$$\text{Out[119]= } 0$$

$$\text{Out[120]= } \frac{4 b}{3 \pi}$$

That $\bar{x} = 0$ is also clear from the fact that the density is a constant and the upper half of the ellipse is symmetric with respect to the y -axis.

The mass of the lamina, the moments of the lamina with respect to the x - and y -axis for various values of a and b are as follows:

```
In[121]:= umassa = TableForm[Table[Mua[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {"a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}];
uxmomenta = TableForm[Table[Mxua[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {"a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}];
uymomenta = TableForm[Table[Myua[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {"a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}];
TableForm[{umassa, uxmomenta, uymomenta},
  TableHeadings -> {"Mass", "x-moment", "y-moment"}, {}]
```

Out[124]//TableForm=

		b=1	b=2	b=3
Mass	a=1	$\frac{\pi}{2}$	π	$\frac{3\pi}{2}$
	a=2	π	2π	3π
	a=3	$\frac{3\pi}{2}$	3π	$\frac{9\pi}{2}$
		b=1	b=2	b=3
x-moment	a=1	$\frac{2}{3}$	$\frac{8}{3}$	6
	a=2	$\frac{4}{3}$	$\frac{16}{3}$	12
	a=3	2	8	18
		b=1	b=2	b=3
y-moment	a=1	0	0	0
	a=2	0	0	0
	a=3	0	0	0

The corresponding y-coordinate of the center of mass in each case is (recall that $\bar{x} = 0$ for all cases)

```
In[125]:= centermassua = Table[ $\frac{Mxua[a, b]}{Mua[a, b]}$ , {a, 1, 3}, {b, 1, 3}];
TableForm[centermassua,
  TableHeadings -> {"a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}]
```

Out[126]//TableForm=

		b=1	b=2	b=3
a=1		$\frac{4}{3\pi}$	$\frac{8}{3\pi}$	$\frac{4}{\pi}$
a=2		$\frac{4}{3\pi}$	$\frac{8}{3\pi}$	$\frac{4}{\pi}$
a=3		$\frac{4}{3\pi}$	$\frac{8}{3\pi}$	$\frac{4}{\pi}$


The following animation shows how the center of mass changes as a and b varies.

```
In[127]:= plot4a[a_, b_] :=
  Plot[f[x, a, b], {x, -a, a}, PlotRange -> {{-5, 5}, {-15, 15}}, Filling -> Axis];
plot5a[a_, b_] := ListPlot[ $\left\{\left\{\frac{Myua[a, b]}{Mua[a, b]}, \frac{Mxua[a, b]}{Mua[a, b]}\right\}\right\}$ ,
  PlotStyle -> {Red, PointSize[0.02]}]
plotua[a_, b_] := Show[plot4a[a, b], plot5a[a, b]]
```

Important Note: If you are reading the printed version of this publication, then you will not be able to view any of the animations generated from the **Animate** command in this chapter. If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then evaluate each **Animate** command to view the corresponding animation. Just click

on the arrow button to start the animation. To control the animation just click at various points on the sliding bar or else manually drag the bar.

```
In[130]:= Animate[plotua[a, b], {a, 1, 8}, {b, 1, 10}]
```



b) Here, $\rho = e^{-x}$. With the above notations modified to reflect variable density, we have

```
In[131]:= Clear[a, b, Mv, Mxv, Myv, xbarv, ybarv]
```

$$\rho[x_] = E^{-x}$$

$$Mvb[a_, b_] = \int_{-a}^a \rho[x] fa[x, a, b] dx$$

$$Mxvb[a_, b_] = (1/2) \int_{-a}^a \rho[x] (fa[x, a, b])^2 dx$$

$$Myvb[a_, b_] = \int_{-a}^a \rho[x] x fa[x, a, b] dx$$

```
Out[132]= e-x
```

```
Out[133]= ConditionalExpression[b π BesselI[1, a], a > 0]
```

```
Out[134]= 
$$\frac{2 b^2 (a \text{Cosh}[a] - \text{Sinh}[a])}{a^2}$$

```

```
Out[135]= ConditionalExpression[-a b π BesselI[2, a], a > 0]
```

$$\text{In[136]:= } \bar{x}[\mathbf{a}, \mathbf{b}] = \frac{\text{Myvb}[\mathbf{a}, \mathbf{b}]}{\text{Mvb}[\mathbf{a}, \mathbf{b}]}$$

$$\bar{y}[\mathbf{a}, \mathbf{b}] = \frac{\text{Mxvb}[\mathbf{a}, \mathbf{b}]}{\text{Mvb}[\mathbf{a}, \mathbf{b}]}$$

$$\text{Out[136]= } \text{ConditionalExpression}\left[-\frac{a \text{ BesselI}[2, a]}{\text{BesselI}[1, a]}, a > 0\right]$$

$$\text{Out[137]= } \text{ConditionalExpression}\left[\frac{2 b (a \text{ Cosh}[a] - \text{Sinh}[a])}{a^2 \pi \text{ BesselI}[1, a]}, a > 0\right]$$

Observe that the formulas for the mass and moments of the lamina are no longer elementary. Here is a table of numerical values for these quantities assuming various choices for a and b :

```
In[138]:= umassb = TableForm[Table[Mvb[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {{ "a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}}];
uxmomentb = TableForm[Table[Mxvb[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {{ "a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}}];
uymomentb = TableForm[Table[Myvb[a, b], {a, 1, 3}, {b, 1, 3}],
  TableHeadings -> {{ "a=1", "a=2", "a=3"}, {"b=1", "b=2", "b=3"}}];
TableForm[N[{umassb, uxmomentb, uymomentb}],
  TableHeadings -> {{ "Mass", "x-moment", "y-moment"}, {}}]
```

Out[141]/TableForm=

		b=1	b=2	b=3
Mass	a=1	1.7755	3.551	5.3265
	a=2	4.99713	9.99427	14.9914
	a=3	12.4199	24.8398	37.2596
x-moment		b=1	b=2	b=3
	a=1	0.735759	2.94304	6.62183
	a=2	1.94877	7.79506	17.5389
y-moment		b=1	b=2	b=3
	a=1	-0.426464	-0.852928	-1.27939
	a=2	-4.32879	-8.65758	-12.9864
	a=3	-21.1606	-42.3213	-63.4819

The coordinates for the center of mass are

```
In[142]:= centermassvb = N[Table[{Myvb[a, b]/Mvb[a, b], Mxvb[a, b]/Mvb[a, b]}, {a, 1, 3}, {b, 1, 3}]]
```

```
Out[142]= {{{-0.240194, 0.414395}, {-0.240194, 0.828791}, {-0.240194, 1.24319}},
  {{-0.866255, 0.389977}, {-0.866255, 0.779953}, {-0.866255, 1.16993}},
  {{-1.70377, 0.361161}, {-1.70377, 0.722323}, {-1.70377, 1.08348}}}
```

Here is a plot showing the two centers of mass with for uniform and variable density.

In[143]:=

```


plot4b[a_, b_] := Plot[f[x, a, b], {x, -a, a},
  PlotRange -> {{-8, 8}, {-1, 8}}, AspectRatio -> Automatic, Filling -> Axis];
plot5b[a_, b_] := ListPlot[{{Myvb[a, b], Mxvb[a, b]},
  {Mvb[a, b], Mvb[a, b]}},
  PlotStyle -> {Green, PointSize[0.02]}]
plotvb[a_, b_] := Show[plot4b[a, b], plot5b[a, b]]

```

Important Note: If you are reading the printed version of this publication, then you will not be able to view any of the animations generated from the **Animate** command in this chapter. If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then evaluate each **Animate** command to view the corresponding animation. Just click on the arrow button to start the animation. To control the animation just click at various points on the sliding bar or else manually drag the bar.

In[146]:= **Animate**[plotvb[a, b], {a, 1, 8}, {b, 1, 8}]

Out[146]=



■ Exercises

- Suppose a lamina is lying underneath the graph of $y = 1 + x^2$ over the interval $[0, 2]$.
 - Assume the density of the lamina is $\rho = 3$. Find the mass, moments, and the center of mass of the lamina.
 - Assume the density of the lamina is $\rho = 2x$. Find the mass, moments, and the center of mass of the lamina.
 - Plot the lamina and the center of mass on the same axes for both parts a) and b) above.

2. Suppose a lamina of constant density $\rho = 2$ is in the shape of the astroid $x^{2/3} + y^{2/3} = 1$. Find its mass, moments, and center of mass. Plot the lamina with its center of mass.